

---

# VeriPass

*Sistem Verifikasi Aset Berbasis Blockchain*

---

Manual Book

Januari 2026

# Daftar Isi

<b>1</b>	<b>Penjelasan Umum</b>	<b>4</b>
1.1	Deskripsi Singkat Sistem . . . . .	4
1.2	Tujuan Manual . . . . .	4
1.3	Gambaran Umum Fungsionalitas . . . . .	4
1.4	Komponen Utama Sistem . . . . .	5
<b>2</b>	<b>Rangka Perangkat Lunak</b>	<b>6</b>
2.1	Struktur Umum Sistem . . . . .	6
2.2	Smart Contract . . . . .	6
2.2.1	AssetPassport . . . . .	6
2.2.2	EventRegistry . . . . .	7
2.3	Oracle . . . . .	7
2.4	Backend . . . . .	9
2.5	Frontend . . . . .	9
2.6	Pattern dan Data Flow . . . . .	10
2.6.1	Data Integrity Pattern . . . . .	10
2.6.2	Authentication Flow . . . . .	10
<b>3</b>	<b>Setup</b>	<b>11</b>
3.1	Persiapan dan Konfigurasi . . . . .	11
3.2	Menjalankan Smart Contract . . . . .	11
3.3	Menjalankan Oracle . . . . .	11
3.4	Menjalankan Backend . . . . .	12
3.5	Menjalankan Frontend . . . . .	12
3.6	Urutan Eksekusi Keseluruhan . . . . .	12
<b>4</b>	<b>Penggunaan</b>	<b>13</b>
4.1	Asset Minting . . . . .	13
4.2	Asset Viewing dan Verification . . . . .	14
4.3	Verified Event . . . . .	15
4.3.1	Custom Event (User-Submitted) . . . . .	15
4.3.2	Oracle-Verified Event . . . . .	15
<b>5</b>	<b>Lain-lain</b>	<b>17</b>
5.1	Catatan Keamanan . . . . .	17
5.1.1	Private Key Management . . . . .	17
5.1.2	Smart Contract Security . . . . .	17
5.1.3	Backend Security . . . . .	17
5.1.4	Data Integrity . . . . .	17
5.2	Kesimpulan dan Saran . . . . .	18
5.2.1	Saran Pengembangan . . . . .	18
<b>A</b>	<b>Source Code Smart Contract</b>	<b>19</b>
A.1	AssetPassport.sol . . . . .	19
A.2	EventRegistry.sol . . . . .	21
<b>B</b>	<b>Database Schema</b>	<b>25</b>

# Daftar Gambar

1.1	Arsitektur Sistem VeriPass . . . . .	5
2.1	Alur Pemrosesan Oracle . . . . .	8
2.2	Alur Data dan Verifikasi . . . . .	10
4.1	Form Minting Passport . . . . .	13
4.2	Detail Passport dengan Verification Badge . . . . .	14
4.3	Event Timeline dengan Filter . . . . .	16

# Daftar Tabel

2.1	Stack Teknologi VeriPass . . . . .	6
2.2	Fungsi Utama AssetPassport . . . . .	7
2.3	Endpoint API Utama . . . . .	9

# Bab 1

## Penjelasan Umum

### 1.1 Deskripsi Singkat Sistem

VeriPass adalah sistem verifikasi aset berbasis blockchain yang menciptakan paspor digital anti-manipulasi untuk aset fisik menggunakan standar ERC-721 NFT. Sistem ini dirancang untuk memberikan bukti kepemilikan dan riwayat aset yang transparan, terdesentralisasi, dan dapat diverifikasi secara kriptografis.

Setiap aset fisik direpresentasikan sebagai NFT unik pada blockchain Ethereum, dengan hash metadata yang tersimpan secara permanen. Integritas data dijamin melalui perbandingan hash on-chain dan off-chain, memastikan setiap modifikasi data dapat terdeteksi.

### 1.2 Tujuan Manual

Dokumen ini bertujuan sebagai panduan teknis komprehensif untuk:

1. Memahami arsitektur dan komponen sistem VeriPass
2. Melakukan instalasi dan konfigurasi sistem secara lengkap
3. Mengoperasikan fitur-fitur utama aplikasi
4. Memahami aspek keamanan dan praktik terbaik penggunaan

### 1.3 Gambaran Umum Fungsionalitas


VeriPass menyediakan fungsionalitas utama berikut:

- **Minting Paspor Aset** – Pembuatan NFT unik untuk setiap aset fisik dengan metadata terverifikasi
- **Verifikasi Integritas** – Perbandingan hash on-chain dan off-chain untuk memastikan data tidak dimanipulasi
- **Pencatatan Event** – Riwayat lengkap lifecycle aset termasuk maintenance, sertifikasi, dan transfer kepemilikan
- **Oracle Integration** – Verifikasi otomatis service record dari provider terpercaya
- **Web3 Authentication** – Autentikasi berbasis wallet untuk keamanan maksimal

## 1.4 Komponen Utama Sistem

Sistem VeriPass terdiri dari empat komponen utama yang saling terintegrasi, sebagaimana diilustrasikan pada Gambar 1.1.

1. **Frontend** – Aplikasi React dengan RainbowKit untuk koneksi wallet dan wagmi untuk interaksi smart contract
2. **Backend** – Server Hono.js dengan PostgreSQL database, menyediakan REST API dan autentikasi JWT
3. **Smart Contracts** – Kontrak Solidity pada blockchain Ethereum (AssetPassport dan EventRegistry)
4. **Oracle** – Worker service yang memproses service record terverifikasi dan mencatatnya ke blockchain



Gambar 1.1: Arsitektur Sistem VeriPass

# Bab 2

## Rangka Perangkat Lunak

### 2.1 Struktur Umum Sistem

VeriPass menggunakan arsitektur three-tier yang memisahkan presentasi, logika bisnis, dan penyimpanan data. Teknologi yang digunakan dirangkum dalam Tabel 2.1.

Tabel 2.1: Stack Teknologi VeriPass

Kategori	Teknologi
Frontend	React, Vite, RainbowKit, wagmi
Backend	Hono.js, Drizzle ORM
Database	PostgreSQL
Blockchain	Solidity, Hardhat, ethers.js
Runtime	Bun

Direktori `shared/` berfungsi sebagai satu-satunya sumber ABI kontrak yang digunakan bersama oleh frontend dan backend. Setelah perubahan kontrak, ABI harus di-export ulang menggunakan script `export-abi.ts`.

### 2.2 Smart Contract

#### 2.2.1 AssetPassport

Kontrak `AssetPassport` mengimplementasikan standar ERC-721 untuk merepresentasikan paspor aset sebagai NFT. Fitur utama meliputi:

- **Minting dengan Metadata Hash** – Setiap passport menyimpan hash keccak256 dari metadata off-chain
- **Authorized Minters** – Hanya alamat terotorisasi atau owner yang dapat melakukan minting
- **Ownership Tracking** – Counter `ownershipHands` melacak berapa kali aset berpindah tangan
- **Emergency Controls** – Fungsi `pause/unpause` untuk kondisi darurat

Fungsi-fungsi utama kontrak dirangkum pada Tabel 2.2.

Tabel 2.2: Fungsi Utama AssetPassport

Fungsi	Deskripsi
<code>mintPassport</code>	Minting passport baru dengan metadata hash
<code>getAssetInfo</code>	Mengambil informasi asset (hash, timestamp, status)
<code>getOwnershipHand</code>	Mengembalikan jumlah perpindahan kepemilikan
<code>addAuthorizedMinter</code>	Menambah alamat sebagai authorized minter
<code>deactivatePassport</code>	Menonaktifkan passport (aset hilang/dicuri)

## 2.2.2 EventRegistry

Kontrak `EventRegistry` berfungsi sebagai append-only log untuk mencatat seluruh event lifecycle aset. Tipe event yang didukung:

- MAINTENANCE (0) – Service, repair, inspeksi
- VERIFICATION (1) – Verifikasi keaslian oleh oracle
- WARRANTY (2) – Klaim dan perpanjangan garansi
- CERTIFICATION (3) – Sertifikasi pihak ketiga
- CUSTOM (4) – Event kustom dari pengguna

Terdapat dua jalur pencatatan event:


1. **User Events** (`recordEvent`) – Disubmit langsung oleh pemilik aset, bersifat unverified
2. **Oracle Events** (`recordVerifiedEvent`) – Disubmit oleh trusted oracle dengan signature, bersifat verified

## 2.3 Oracle

Oracle adalah background worker yang menjembatani data off-chain dengan on-chain. Proses kerja oracle:

1. Polling database setiap 15-30 detik untuk service record baru
2. Memvalidasi record yang sudah diverifikasi oleh provider
3. Menghitung hash deterministik dari data event
4. Menandatangani hash dengan wallet oracle
5. Submit ke kontrak EventRegistry sebagai verified event
6. Update status record: PENDING → PROCESSING → COMPLETED/FAILED





Gambar 2.1: Alur Pemrosesan Oracle

## 2.4 Backend

Backend dibangun dengan Hono.js dan menyediakan REST API untuk operasi berikut:

- **Authentication** – Web3 signature-based auth (nonce → sign → verify → JWT)
- **Asset Management** – CRUD operasi untuk metadata aset
- **Evidence Tracking** – Pencatatan dan query event/evidence
- **Provider Integration** – API untuk service provider eksternal

Endpoint utama API dirangkum pada Tabel 2.3.

Tabel 2.3: Endpoint API Utama

Method	Endpoint	Deskripsi
POST	/api/auth/nonce	Request nonce untuk signing
POST	/api/auth/verify	Verifikasi signature, dapatkan JWT
POST	/api/assets	Buat asset baru
GET	/api/assets/:id	Query asset by ID
POST	/api/evidence	Catat evidence baru
GET	/api/evidence/asset/:id	Query evidence by asset

## 2.5 Frontend

Frontend React menggunakan arsitektur berbasis hooks dengan pemisahan jelas antara API hooks dan contract hooks:

- **/hooks/api/** – TanStack Query hooks untuk komunikasi dengan backend
- **/hooks/contracts/** – wagmi hooks untuk interaksi smart contract

Halaman utama aplikasi:

- **HomePage** – Landing page dengan informasi sistem
- **PassportsPage** – Daftar semua passport yang di-mint
- **MintPage** – Form untuk minting passport baru
- **PassportDetailsPage** – Detail passport dengan verification dan event timeline


Frontend mendukung **graceful degradation** – tetap beroperasi dalam mode offline ketika backend tidak tersedia, dengan fitur terbatas pada operasi blockchain langsung.

## 2.6 Pattern dan Data Flow

### 2.6.1 Data Integrity Pattern

Integritas data dijamin melalui hash kriptografis:

1. Metadata di-hash menggunakan keccak256 dengan key sorting deterministik
2. Hash disimpan on-chain saat minting
3. Verifikasi dilakukan dengan membandingkan hash on-chain vs computed hash dari data off-chain
4. Mismatch mengindikasikan data telah dimodifikasi



eraser

Gambar 2.2: Alur Data dan Verifikasi

### 2.6.2 Authentication Flow

1. User menghubungkan wallet via RainbowKit
2. Frontend request nonce dari backend
3. User menandatangani message berisi nonce
4. Backend memverifikasi signature, issue JWT token (valid 7 hari)
5. Token digunakan untuk autentikasi request selanjutnya

# Bab 3

## Setup

### 3.1 Persiapan dan Konfigurasi

Prasyarat sistem yang diperlukan:

- Node.js v18+ atau Bun runtime
- PostgreSQL database
- Git untuk version control
- MetaMask atau wallet Web3 lainnya

Clone repository dan install dependencies:

```
1 git clone <repository-url>
2 cd veripass
```

### 3.2 Menjalankan Smart Contract

Navigasi ke direktori contracts dan jalankan local blockchain:

```
1 cd contracts
2 bun install
3 bun hardhat compile           # Compile kontrak
4 bun hardhat node              # Jalankan local node (port 8545)
```

Pada terminal terpisah, deploy kontrak:

```
1 bun hardhat run scripts/deploy.ts --network localhost
2 bun hardhat run scripts/export-abi.ts # Export ABI ke shared/
```

Konfigurasi MetaMask untuk testing lokal:

- RPC URL: `http://127.0.0.1:8545`
- Chain ID: 31337
- Test private key tersedia dari output Hardhat node

### 3.3 Menjalankan Oracle

Oracle memerlukan konfigurasi environment variable:

```
1 # Di file backend/.env
2 ORACLE_API_KEY=<min-32-char-key>
3 ORACLE_PRIVATE_KEY=0x<64-char-hex>
4 POLL_INTERVAL=30000
5 EVENT_REGISTRY_ADDRESS=<deployed-address>
```

Jalankan oracle worker:

```
1 cd backend
2 bun run dev:oracle
```

Oracle akan mendaftarkan alamatnya dan mulai polling untuk service record baru.

## 3.4 Menjalankan Backend

Setup database dan jalankan server:

```
1 cd backend
2 cp .env.example .env           # Salin dan edit konfigurasi
3 bun install
4 bun run db:push                # Push schema ke database
5 bun run db:seed                # (Opsional) Seed test data
6 bun run dev                    # Jalankan server (port 3000)
```

Environment variables yang diperlukan:

```
1 DB_URL=postgresql://user:pass@localhost:5432/veripass
2 JWT_SECRET=<min-32-char-secret>
3 FRONTEND_URL=http://localhost:5173
4 ASSET_PASSPORT_ADDRESS=<deployed-address>
5 EVENT_REGISTRY_ADDRESS=<deployed-address>
```

## 3.5 Menjalankan Frontend

Setup dan jalankan development server:

```
1 cd frontend
2 cp .env.example .env           # Salin dan edit konfigurasi
3 bun install
4 bun run dev                    # Jalankan server (port 5173)
```

Environment variables frontend:

```
1 VITE_API_URL=http://localhost:3000
2 VITE_WALLETCONNECT_PROJECT_ID=<project-id>
3 VITE_ASSET_PASSPORT_ADDRESS=<deployed-address>
4 VITE_EVENT_REGISTRY_ADDRESS=<deployed-address>
```

## 3.6 Urutan Eksekusi Keseluruhan

Untuk menjalankan sistem lengkap, ikuti urutan berikut:

1. **Database** – Pastikan PostgreSQL berjalan
2. **Blockchain** – Jalankan Hardhat node, deploy kontrak
3. **Backend** – Push schema, jalankan server
4. **Oracle** – Jalankan worker (opsional, untuk verified events)
5. **Frontend** – Jalankan dev server

Verifikasi instalasi dengan mengakses `http://localhost:5173` dan menghubungkan wallet.


# Bab 4

## Penggunaan

### 4.1 Asset Minting

Proses minting passport aset baru:

1. Hubungkan wallet melalui tombol Connect pada header
2. Navigasi ke halaman Mint
3. Isi form dengan data aset:
  - Alamat penerima (recipient address)
  - Manufacturer dan model
  - Serial number
  - Tanggal manufaktur
  - Deskripsi (opsional)
4. Klik tombol Mint
5. Konfirmasi transaksi di wallet
6. Tunggu konfirmasi blockchain



Gambar 4.1: Form Minting Passport


Sistem akan:

- Menghitung hash deterministik dari metadata
- Menyimpan metadata ke database dengan status PENDING
- Submit transaksi minting ke blockchain
- Update status ke MINTED setelah konfirmasi

## 4.2 Asset Viewing dan Verification

Melihat dan memverifikasi passport:

1. Navigasi ke halaman Passports untuk melihat daftar semua passport
2. Klik passport untuk melihat detail
3. Perhatikan **Verification Badge**:
  - **Verified** – Hash on-chain cocok dengan off-chain
  - **Pending** – Menunggu konfirmasi
  - **Mismatch** – Hash tidak cocok, data mungkin dimodifikasi



Gambar 4.2: Detail Passport dengan Verification Badge

Informasi yang ditampilkan pada halaman detail:

- Metadata aset (manufacturer, model, serial number)
- Status verifikasi dengan perbandingan hash
- Owner address saat ini
- Ownership hands (jumlah perpindahan)
- Timestamp minting
- Event timeline

## 4.3 Verified Event

Terdapat dua cara mencatat event pada passport:

### 4.3.1 Custom Event (User-Submitted)

Event yang dicatat langsung oleh pemilik aset:

1. Buka halaman detail passport
2. Klik tombol “Record Event”
3. Isi form event:
  - Tipe event (Custom)
  - Deskripsi
  - Tanggal event
  - Data tambahan (JSON)
4. Konfirmasi transaksi di wallet

Custom event bersifat **unverified** dan ditandai pada timeline.

### 4.3.2 Oracle-Verified Event


Event yang diverifikasi oleh oracle dari service provider:

1. Service provider submit record via Provider API
2. Oracle secara otomatis polling record baru
3. Oracle memverifikasi dan menandatangani data
4. Event dicatat ke blockchain sebagai verified
5. Event muncul di timeline dengan badge verified

Event timeline mendukung filtering berdasarkan tipe:

- Maintenance
- Verification
- Warranty
- Certification
- Custom





Gambar 4.3: Event Timeline dengan Filter

# Bab 5

## Lain-lain

### 5.1 Catatan Keamanan

Aspek keamanan yang perlu diperhatikan:

#### 5.1.1 Private Key Management

- **Jangan pernah** membagikan private key wallet
- Gunakan hardware wallet untuk aset bernilai tinggi
- Backup seed phrase di lokasi aman dan offline

#### 5.1.2 Smart Contract Security

- Kontrak menggunakan OpenZeppelin library yang sudah diaudit
- Implementasi Pausable untuk emergency stop
- Access control dengan Ownable dan authorized minters pattern
- Input validation untuk mencegah zero address dan invalid hash

#### 5.1.3 Backend Security

- JWT secret minimal 32 karakter
- Nonce expiry 5 menit untuk mencegah replay attack
- CORS dikonfigurasi hanya untuk frontend URL
- API key hashing untuk provider authentication

#### 5.1.4 Data Integrity

- Selalu verifikasi hash on-chain sebelum mempercayai data off-chain
- Perhatikan warning jika terdapat hash mismatch
- Oracle signature memastikan keaslian verified events

## 5.2 Kesimpulan dan Saran

VeriPass menyediakan solusi verifikasi aset yang transparan dan tamper-proof dengan memanfaatkan teknologi blockchain. Sistem ini cocok untuk:

- Tracking aset bernilai tinggi (kendaraan, elektronik, barang koleksi)
- Sertifikasi dan maintenance record
- Proof of ownership dan transfer history
- Audit trail yang tidak dapat dimanipulasi

### 5.2.1 Saran Pengembangan

1. **Multi-chain Support** – Deploy ke multiple blockchain untuk redundansi
2. **IPFS Integration** – Simpan metadata lengkap di IPFS untuk desentralisasi penuh
3. **Mobile App** – Aplikasi mobile untuk scanning dan verifikasi cepat
4. **Provider Network** – Ekspansi jaringan service provider terverifikasi

# Lampiran A

## Source Code Smart Contract

### A.1 AssetPassport.sol

```
1 // SPDX-License-Identifier: MIT
2 pragma solidity ^0.8.28;
3
4 import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
5 import "@openzeppelin/contracts/access/Ownable.sol";
6 import "@openzeppelin/contracts/utils/Pausable.sol";
7 import "../interfaces/IAssetPassport.sol";
8 import "../errors/VeriPassErrors.sol";
9
10 contract AssetPassport is ERC721, Ownable, Pausable, IAssetPassport {
11     uint256 private _tokenIdCounter;
12     mapping(uint256 => AssetInfo) private _assets;
13     mapping(address => bool) private _authorizedMinters;
14     mapping(uint256 => uint32) private _ownershipHands;
15
16     constructor()
17         ERC721("VeriPass Asset Passport", "VPASS")
18         Ownable(msg.sender)
19     {
20         _tokenIdCounter = 1;
21     }
22
23     function mintPassport(
24         address to,
25         bytes32 metadataHash
26     ) external override whenNotPaused returns (uint256 tokenId) {
27         if (!_authorizedMinters[msg.sender] && msg.sender != owner()) {
28             revert VeriPass_AssetPassport_NotAuthorizedMinter();
29         }
30         if (to == address(0)) revert VeriPass_ZeroAddress();
31         if (metadataHash == bytes32(0))
32             revert VeriPass_AssetPassport_InvalidMetadataHash();
33         tokenId = _tokenIdCounter++;
34         _safeMint(to, tokenId);
35         _assets[tokenId] = AssetInfo({
36             metadataHash: metadataHash,
37             mintTimestamp: uint40(block.timestamp),
38             isActive: true
39         });
40
41         emit PassportMinted(tokenId, to, metadataHash);
42
43         return tokenId;
44     }
45
46     function getAssetInfo(
47         uint256 tokenId
48     ) external view override returns (AssetInfo memory info) {
49         if (_ownerOf(tokenId) == address(0)) {
50             revert VeriPass_AssetPassport_TokenDoesNotExist(tokenId);
51         }
52     }
53 }
```

```

51     }
52     return _assets[tokenId];
53 }
54
55 function getOwnershipHand(
56     uint256 tokenId
57 ) external view override returns (uint32 handCount) {
58     if (_ownerOf(tokenId) == address(0)) {
59         revert VeriPass_AssetPassport_TokenDoesNotExist(tokenId);
60     }
61     return _ownershipHands[tokenId];
62 }
63
64 function isAuthorizedMinter(
65     address account
66 ) external view override returns (bool isAuthorized) {
67     return _authorizedMinters[account] || account == owner();
68 }
69
70 function nextTokenId() external view returns (uint256) {
71     return _tokenIdCounter;
72 }
73
74 function addAuthorizedMinter(address minter) external override onlyOwner {
75     if (minter == address(0)) revert VeriPass_ZeroAddress();
76     _authorizedMinters[minter] = true;
77     emit MinterUpdated(minter, true);
78 }
79
80 function removeAuthorizedMinter(
81     address minter
82 ) external override onlyOwner {
83     _authorizedMinters[minter] = false;
84     emit MinterUpdated(minter, false);
85 }
86
87 function deactivatePassport(uint256 tokenId) external override onlyOwner {
88     if (_ownerOf(tokenId) == address(0)) {
89         revert VeriPass_AssetPassport_TokenDoesNotExist(tokenId);
90     }
91     _assets[tokenId].isActive = false;
92     emit PassportDeactivated(tokenId);
93 }
94
95 function pause() external override onlyOwner {
96     _pause();
97 }
98
99 function unpause() external override onlyOwner {
100     _unpause();
101 }
102
103 function _update(
104     address to,
105     uint256 tokenId,
106     address auth
107 ) internal override whenNotPaused returns (address previousOwner) {
108     previousOwner = super._update(to, tokenId, auth);
109
110     if (previousOwner == address(0)) {

```

```

111         _ownershipHands[tokenId] = 1;
112     } else if (previousOwner != to) {
113         unchecked {
114             _ownershipHands[tokenId] += 1;
115         }
116     }
117
118     return previousOwner;
119 }
120 }

```

Listing A.1: AssetPassport.sol

## A.2 EventRegistry.sol

```

1 // SPDX-License-Identifier: MIT
2 pragma solidity ^0.8.28;
3
4 import "@openzeppelin/contracts/access/Ownable.sol";
5 import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
6 import "../interfaces/IEventRegistry.sol";
7 import "../interfaces/IAssetPassport.sol";
8 import "../errors/VeriPassErrors.sol";
9
10 contract EventRegistry is Ownable, IEventRegistry {
11     uint256 private _eventIdCounter;
12
13     mapping(uint256 => uint256[]) private _assetEvents;
14
15     mapping(uint256 => LifecycleEvent) private _events;
16
17     mapping(address => bool) private _trustedOracles;
18
19     /// inject the asset passport contract
20     IAssetPassport public assetPassport;
21
22     constructor() Ownable(msg.sender) {
23         _eventIdCounter = 1;
24     }
25
26     function recordEvent(
27         uint256 assetId,
28         EventType eventType,
29         bytes32 dataHash
30     ) external override returns (uint256 eventId) {
31         if (dataHash == bytes32(0))
32             revert VeriPass_EventRegistry_InvalidDataHash();
33
34         _validateEventSubmission(assetId, msg.sender);
35
36         eventId = _eventIdCounter++;
37
38         _events[eventId] = LifecycleEvent({
39             id: eventId,
40             assetId: assetId,
41             eventType: eventType,
42             submitter: msg.sender,
43             timestamp: uint40(block.timestamp),
44             dataHash: dataHash
45         });

```

```

46     _assetEvents[assetId].push(eventId);
47
48     emit EventRecorded(assetId, eventId, eventType, msg.sender, dataHash);
49
50     return eventId;
51 }
52
53
54 function recordVerifiedEvent (
55     uint256 assetId,
56     EventType eventType,
57     bytes32 dataHash,
58     bytes calldata oracleSignature
59 ) external override returns (uint256 eventId) {
60     if (!_trustedOracles[msg.sender]) {
61         revert VeriPass_EventRegistry_NotTrustedOracle(msg.sender);
62     }
63
64     if (dataHash == bytes32(0))
65         revert VeriPass_EventRegistry_InvalidDataHash();
66     if (oracleSignature.length == 0) {
67         revert VeriPass_EventRegistry_InvalidOracleSignature();
68     }
69
70     // Only allow verified event types (not CUSTOM - that's for user-
71     submitted events)
72     if (eventType == EventType.CUSTOM) {
73         revert VeriPass_EventRegistry_InvalidEventType();
74     }
75
76     // verify that asset exists
77     if (address(assetPassport) != address(0)) {
78         try assetPassport.getAssetInfo(assetId) returns (
79             AssetInfo memory
80         ) {} catch {
81             revert VeriPass_EventRegistry_AssetNotFound(assetId);
82         }
83     }
84
85     eventId = _eventIdCounter++;
86
87     _events[eventId] = LifecycleEvent({
88         id: eventId,
89         assetId: assetId,
90         eventType: eventType,
91         submitter: msg.sender,
92         timestamp: uint40(block.timestamp),
93         dataHash: dataHash
94     });
95
96     _assetEvents[assetId].push(eventId);
97
98     emit EventRecorded(
99         assetId,
100         eventId,
101         eventType,
102         msg.sender,
103         dataHash
104     );

```

```

105     return eventId;
106 }
107
108 function getEventsByAsset (
109     uint256 assetId
110 ) external view override returns (LifecycleEvent[] memory events) {
111     uint256[] memory eventIds = _assetEvents[assetId];
112     events = new LifecycleEvent[] (eventIds.length);
113
114     for (uint256 i = 0; i < eventIds.length; i++) {
115         events[i] = _events[eventIds[i]];
116     }
117
118     return events;
119 }
120
121 function getEventsByType (
122     uint256 assetId,
123     EventType eventType
124 ) external view override returns (LifecycleEvent[] memory events) {
125     uint256[] memory eventIds = _assetEvents[assetId];
126
127     uint256 matchCount = 0;
128     for (uint256 i = 0; i < eventIds.length; i++) {
129         if (_events[eventIds[i]].eventType == eventType) {
130             matchCount++;
131         }
132     }
133
134     events = new LifecycleEvent[] (matchCount);
135     uint256 currentIndex = 0;
136
137     for (uint256 i = 0; i < eventIds.length; i++) {
138         if (_events[eventIds[i]].eventType == eventType) {
139             events[currentIndex] = _events[eventIds[i]];
140             currentIndex++;
141         }
142     }
143
144     return events;
145 }
146
147 function getEventCount (
148     uint256 assetId
149 ) external view override returns (uint256 count) {
150     return _assetEvents[assetId].length;
151 }
152
153 function getEvent (
154     uint256 eventId
155 ) external view returns (LifecycleEvent memory lifecycleEvent) {
156     if (_events[eventId].id == 0) {
157         revert VeriPass_EventRegistry_EventNotFound(eventId);
158     }
159     return _events[eventId];
160 }
161
162 function setAssetPassport(address passportAddress) external onlyOwner {
163     if (passportAddress == address(0)) revert VeriPass_ZeroAddress();
164     assetPassport = IAssetPassport(passportAddress);

```



```

165     }
166
167     function addTrustedOracle(address oracle) external override onlyOwner {
168         if (oracle == address(0)) revert VeriPass_ZeroAddress();
169         if (_trustedOracles[oracle]) {
170             revert VeriPass_EventRegistry_OracleAlreadyRegistered(oracle);
171         }
172
173         _trustedOracles[oracle] = true;
174         emit OracleAdded(oracle);
175     }
176
177     function removeTrustedOracle(address oracle) external override onlyOwner {
178         if (!_trustedOracles[oracle]) {
179             revert VeriPass_EventRegistry_OracleNotRegistered(oracle);
180         }
181
182         _trustedOracles[oracle] = false;
183         emit OracleRemoved(oracle);
184     }
185
186     function isTrustedOracle(
187         address account
188     ) external view override returns (bool isTrusted) {
189         return _trustedOracles[account];
190     }
191
192     function _validateEventSubmission(
193         uint256 assetId,
194         address submitter
195     ) internal view {
196         // AssetPassport must be configured
197         if (address(assetPassport) == address(0)) {
198             revert VeriPass_EventRegistry_AssetNotFound(assetId);
199         }
200
201         // Check asset exists and is active
202         AssetInfo memory info;
203         try assetPassport.getAssetInfo(assetId) returns (
204             AssetInfo memory _info
205         ) {
206             info = _info;
207         } catch {
208             revert VeriPass_EventRegistry_AssetNotFound(assetId);
209         }
210
211         if (!info.isActive) {
212             revert VeriPass_AssetPassport_PassportInactive(assetId);
213         }
214
215         // Check submitter is the asset owner using IERC721.ownerOf
216         address assetOwner = IERC721(address(assetPassport)).ownerOf(assetId);
217         if (submitter != assetOwner) {
218             revert VeriPass_EventRegistry_NotAssetOwner(assetId, submitter);
219         }
220     }
221 }

```

Listing A.2: EventRegistry.sol

# Lampiran B

## Database Schema

```
1 import {
2   pgTable,
3   serial,
4   varchar,
5   timestamp,
6   text,
7   bigint,
8   boolean,
9   jsonb,
10  index,
11  foreignKey,
12 } from "drizzle-orm/pg-core";
13
14 // =====
15 // ASSETS TABLE
16 // =====
17 // Note: 'id' is the database primary key (auto-increment for internal use)
18 //       'assetId' is the blockchain asset ID that matches the smart contract
19
20 export const assets = pgTable(
21   "assets",
22   {
23     id: serial("id").primaryKey(),
24
25     // Must match smart contract
26     assetId: bigint("asset_id", { mode: "number" }).notNull().unique(),
27     dataHash: varchar("data_hash", { length: 66 }).notNull().unique(),
28
29     // Asset metadata (gets hashed)
30     manufacturer: varchar("manufacturer", { length: 255 }).notNull(),
31     model: varchar("model", { length: 255 }).notNull(),
32     serialNumber: varchar("serial_number", { length: 255 }).notNull(),
33     manufacturedDate: varchar("manufactured_date", { length: 10 }), // YYYY-MM-DD
34     description: text("description"),
35
36     // Additional data (flexible)
37     images: jsonb("images").$type<string[]>(), // Array of image URLs
38     metadata: jsonb("metadata").$type<Record<string, unknown>>(), // Any extra fields
39
40     // Mint status: PENDING (waiting for tx), MINTED (confirmed), FAILED (tx failed/cancelled)
41     mintStatus: varchar("mint_status", { length: 20 }).default("PENDING").notNull(),
42     txHash: varchar("tx_hash", { length: 66 }), // Transaction hash when minted
43
44     // Tracking
45     createdBy: varchar("created_by", { length: 42 }).notNull(), // User wallet address
46     createdAt: timestamp("created_at").defaultNow().notNull(),
47     mintedAt: timestamp("minted_at"),
48   },
```

```

49   (table) => [
50     index("asset_id_idx").on(table.assetId),
51     index("data_hash_idx").on(table.dataHash),
52     index("mint_status_idx").on(table.mintStatus),
53   ]
54 );
55
56 // =====
57 // SERVICE PROVIDERS
58 // =====
59
60 export const serviceProviders = pgTable(
61   "service_providers",
62   {
63     id: serial("id").primaryKey(),
64     providerId: varchar("provider_id", { length: 255 }).notNull().unique(),
65     providerName: varchar("provider_name", { length: 255 }).notNull(),
66     providerType: varchar("provider_type", { length: 50 }).notNull(), //
67     manufacturer, service_center, inspector
68     isTrusted: boolean("is_trusted").default(true).notNull(),
69     // API key for provider authentication (hashed)
70     apiKeyHash: varchar("api_key_hash", { length: 128 }),
71     createdAt: timestamp("created_at").defaultNow().notNull(),
72   },
73   (table) => [index("service_providers_provider_id_idx").on(table.providerId)]
74 );
75
76 // =====
77 // SERVICE RECORDS (Provider-submitted records)
78 // =====
79 // Note: 'id' is the database primary key (auto-increment for internal use)
80 //       'recordId' is the business identifier (e.g., "SR-001")
81 // This table receives records from external service providers via API
82
83 export const serviceRecordsProviderA = pgTable(
84   "service_records_provider_a",
85   {
86     id: serial("id").primaryKey(),
87     recordId: varchar("record_id", { length: 255 }).notNull().unique(),
88
89     assetId: bigint("asset_id", { mode: "number" }).notNull(),
90     providerId: varchar("provider_id", { length: 255 }).notNull(),
91
92     // Event classification (provider specifies directly)
93     eventType: varchar("event_type", { length: 20 }).notNull(), // MAINTENANCE,
94     VERIFICATION, WARRANTY, CERTIFICATION
95     eventName: varchar("event_name", { length: 255 }).notNull(), // "Service RAM
96     ", "Battery Replacement", etc.
97
98     // Service details
99     serviceDate: varchar("service_date", { length: 10 }).notNull(), // YYYY-MM-
100     DD
101     technicianName: varchar("technician_name", { length: 255 }),
102     technicianNotes: text("technician_notes"),
103
104     // Work details
105     workPerformed: jsonb("work_performed").$type<string[]>(),
106     partsReplaced: jsonb("parts_replaced").$type<
107       Array<{ name: string; partNumber?: string; quantity?: number }>
108     >(),

```

```

105
106 // Verification status from provider
107 verified: boolean("verified").default(true).notNull(),
108
109 createdAt: timestamp("created_at").defaultNow().notNull(),
110 updatedAt: timestamp("updated_at"),
111 },
112 (table) => [
113   index("service_records_provider_a_asset_id_idx").on(table.assetId),
114   index("service_records_provider_a_provider_id_idx").on(table.providerId),
115   foreignKey({
116     columns: [table.assetId],
117     foreignColumns: [assets.assetId],
118     name: "service_records_provider_a_asset_id_fk",
119   })
120   .onDelete("cascade")
121   .onUpdate("cascade"),
122 ]
123 );
124
125 // =====
126 // PROCESSED SERVICE RECORDS (Oracle processing queue)
127 // =====
128 // Tracks which service records have been processed by the oracle
129
130 export const processedServiceRecords = pgTable(
131   "processed_service_records",
132   {
133     id: serial("id").primaryKey(),
134     serviceRecordId: bigint("service_record_id", { mode: "number" }).notNull(),
135     providerId: varchar("provider_id", { length: 255 }).notNull(),
136
137     // Status: PENDING, PROCESSING, COMPLETED, FAILED
138     status: varchar("status", { length: 20 }).default("PENDING").notNull(),
139
140     // Result (populated after processing)
141     evidenceId: bigint("evidence_id", { mode: "number" }),
142     blockchainEventId: bigint("blockchain_event_id", { mode: "number" }),
143     txHash: varchar("tx_hash", { length: 66 }),
144     errorMessage: text("error_message"),
145
146     createdAt: timestamp("created_at").defaultNow().notNull(),
147     processedAt: timestamp("processed_at"),
148   },
149   (table) => [
150     index("processed_service_records_status_idx").on(table.status),
151     index("processed_service_records_service_record_id_idx").on(
152       table.serviceRecordId
153     ),
154     foreignKey({
155       columns: [table.serviceRecordId],
156       foreignColumns: [serviceRecordsProviderA.id],
157       name: "processed_service_records_service_record_id_fk",
158     })
159     .onDelete("cascade")
160     .onUpdate("cascade"),
161   ]
162 );
163
164 // =====

```

```

165 // EVIDENCE TABLE (Event data)
166 // =====
167
168 export const evidence = pgTable(
169   "evidence",
170   {
171     id: serial("id").primaryKey(),
172
173     // Link to asset
174     assetId: bigint("asset_id", { mode: "number" }).notNull(),
175     dataHash: varchar("data_hash", { length: 66 }).notNull().unique(),
176
177     // Link to service record (for oracle-created evidence)
178     serviceRecordId: bigint("service_record_id", { mode: "number" }),
179
180     // Event details
181     eventType: varchar("event_type", { length: 20 }).notNull(), // MAINTENANCE,
182     // VERIFICATION, WARRANTY, CERTIFICATION, CUSTOM
183     eventDate: varchar("event_date", { length: 10 }), // YYYY-MM-DD
184     providerName: varchar("provider_name", { length: 255 }),
185     description: text("description"),
186
187     // Raw event data (user-provided JSON)
188     eventData: jsonb("event_data").$type<Record<string, unknown>>(),
189
190     // Status: PENDING (oracle flow, not on-chain yet), CONFIRMED (recorded on-
191     // chain)
192     status: varchar("status", { length: 20 }).default("PENDING").notNull(),
193
194     // Verification tracking (for oracle-verified events)
195     isVerified: boolean("is_verified").default(false).notNull(),
196     verifiedBy: varchar("verified_by", { length: 42 }), // Oracle address
197     blockchainEventId: bigint("blockchain_event_id", { mode: "number" }),
198     txHash: varchar("tx_hash", { length: 66 }),
199
200     // Tracking
201     createdBy: varchar("created_by", { length: 42 }).notNull(), // User wallet
202     // address or oracle address
203     createdAt: timestamp("created_at").defaultNow().notNull(),
204     confirmedAt: timestamp("confirmed_at"),
205     verifiedAt: timestamp("verified_at"),
206   },
207   (table) => [
208     index("evidence_asset_id_idx").on(table.assetId),
209     index("evidence_data_hash_idx").on(table.dataHash),
210     index("evidence_status_idx").on(table.status),
211     index("evidence_service_record_id_idx").on(table.serviceRecordId),
212     foreignKey({
213       columns: [table.assetId],
214       foreignColumns: [assets.assetId],
215       name: "evidence_asset_id_fk",
216     })
217       .onDelete("cascade")
218       .onUpdate("cascade"),
219     foreignKey({
220       columns: [table.serviceRecordId],
221       foreignColumns: [serviceRecordsProviderA.id],
222       name: "evidence_service_record_id_fk",
223     })
224       .onDelete("set null")

```

```

222     .onUpdate("cascade"),
223   ]
224 );
225
226 // =====
227 // AUTH NONCES (Web3 authentication)
228 // =====
229
230 export const authNonces = pgTable(
231   "auth_nonces",
232   {
233     id: serial("id").primaryKey(),
234     address: varchar("address", { length: 42 }).notNull().unique(),
235     nonce: varchar("nonce", { length: 66 }).notNull(),
236     expiresAt: timestamp("expires_at").notNull(),
237     createdAt: timestamp("created_at").defaultNow().notNull(),
238   },
239   (table) => [index("auth_nonces_address_idx").on(table.address)]
240 );
241
242 // =====
243 // TYPE EXPORTS
244 // =====
245
246 export type Asset = typeof assets.$inferSelect;
247 export type NewAsset = typeof assets.$inferInsert;
248
249 export type Evidence = typeof evidence.$inferSelect;
250 export type NewEvidence = typeof evidence.$inferInsert;
251
252 export type ServiceProvider = typeof serviceProviders.$inferSelect;
253 export type NewServiceProvider = typeof serviceProviders.$inferInsert;
254
255 export type ServiceRecordProviderA = typeof serviceRecordsProviderA.$inferSelect
256 ;
257 export type NewServiceRecordProviderA =
258   typeof serviceRecordsProviderA.$inferInsert;
259
260 export type ProcessedServiceRecord = typeof processedServiceRecords.$inferSelect
261 ;
262 export type NewProcessedServiceRecord =
263   typeof processedServiceRecords.$inferInsert;
264
265 export type AuthNonce = typeof authNonces.$inferSelect;
266 export type NewAuthNonce = typeof authNonces.$inferInsert;

```

Listing B.1: Database Schema (Drizzle ORM)